

## NAG Library Routine Document

### F08XEF (DHGEQZ)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F08XEF (DHGEQZ) implements the  $QZ$  method for finding generalized eigenvalues of the real matrix pair  $(A, B)$  of order  $n$ , which is in the generalized upper Hessenberg form.

#### 2 Specification

```

SUBROUTINE F08XEF (JOB, COMPQ, COMPZ, N, ILO, IHI, A, LDA, B, LDB,      &
                  ALPHAR, ALPHAI, BETA, Q, LDQ, Z, LDZ, WORK, LWORK,  &
                  INFO)
INTEGER              N, ILO, IHI, LDA, LDB, LDQ, LDZ, LWORK, INFO
REAL (KIND=nag_wp)  A(LDA,*), B(LDB,*), ALPHAR(N), ALPHAI(N), BETA(N), &
                  Q(LDQ,*), Z(LDZ,*), WORK(max(1,LWORK))
CHARACTER(1)        JOB, COMPQ, COMPZ

```

The routine may be called by its LAPACK name *dhgeqz*.

#### 3 Description

F08XEF (DHGEQZ) implements a single-double-shift version of the  $QZ$  method for finding the generalized eigenvalues of the real matrix pair  $(A, B)$  which is in the generalized upper Hessenberg form. If the matrix pair  $(A, B)$  is not in the generalized upper Hessenberg form, then the routine F08WEF (DGGHRD) should be called before invoking F08XEF (DHGEQZ).

This problem is mathematically equivalent to solving the equation

$$\det(A - \lambda B) = 0.$$

Note that, to avoid underflow, overflow and other arithmetic problems, the generalized eigenvalues  $\lambda_j$  are never computed explicitly by this routine but defined as ratios between two computed values,  $\alpha_j$  and  $\beta_j$ :

$$\lambda_j = \alpha_j / \beta_j.$$

The parameters  $\alpha_j$ , in general, are finite complex values and  $\beta_j$  are finite real non-negative values.

If desired, the matrix pair  $(A, B)$  may be reduced to generalized Schur form. That is, the transformed matrix  $B$  is upper triangular and the transformed matrix  $A$  is block upper triangular, where the diagonal blocks are either 1 by 1 or 2 by 2. The 1 by 1 blocks provide generalized eigenvalues which are real and the 2 by 2 blocks give complex generalized eigenvalues.

The parameter JOB specifies two options. If JOB = 'S' then the matrix pair  $(A, B)$  is simultaneously reduced to Schur form by applying one orthogonal transformation (usually called  $Q$ ) on the left and another (usually called  $Z$ ) on the right. That is,

$$\begin{aligned} A &\leftarrow Q^T A Z \\ B &\leftarrow Q^T B Z \end{aligned}$$

The 2 by 2 upper-triangular diagonal blocks of  $B$  corresponding to 2 by 2 blocks of  $A$  will be reduced to non-negative diagonal matrices. That is, if  $A(j+1, j)$  is nonzero, then  $B(j+1, j) = B(j, j+1) = 0$  and  $B(j, j)$  and  $B(j+1, j+1)$  will be non-negative.

If JOB = 'E', then at each iteration the same transformations are computed but they are only applied to those parts of  $A$  and  $B$  which are needed to compute  $\alpha$  and  $\beta$ . This option could be used if generalized eigenvalues are required but not generalized eigenvectors.

If JOB = 'S' and COMPQ = 'V' or 'I', and COMPZ = 'V' or 'I', then the orthogonal transformations used to reduce the pair  $(A, B)$  are accumulated into the input arrays Q and Z. If generalized eigenvectors are required then JOB must be set to JOB = 'S' and if left (right) generalized eigenvectors are to be computed then COMPQ (COMPZ) must be set to COMPQ = 'V' or 'I' and not COMPQ  $\neq$  'N'.

If COMPQ = 'I', then eigenvectors are accumulated on the identity matrix and on exit the array Q contains the left eigenvector matrix  $Q$ . However, if COMPQ = 'V' then the transformations are accumulated on the user-supplied matrix  $Q_0$  in array Q on entry and thus on exit Q contains the matrix product  $QQ_0$ . A similar convention is used for COMPZ.

## 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Moler C B and Stewart G W (1973) An algorithm for generalized matrix eigenproblems *SIAM J. Numer. Anal.* **10** 241–256

Stewart G W and Sun J-G (1990) *Matrix Perturbation Theory* Academic Press, London

## 5 Parameters

- 1: JOB – CHARACTER(1) *Input*  
*On entry:* specifies the operations to be performed on  $(A, B)$ .  
 JOB = 'E'  
 The matrix pair  $(A, B)$  on exit might not be in the generalized Schur form.  
 JOB = 'S'  
 The matrix pair  $(A, B)$  on exit will be in the generalized Schur form.  
*Constraint:* JOB = 'E' or 'S'.
- 2: COMPQ – CHARACTER(1) *Input*  
*On entry:* specifies the operations to be performed on  $Q$ :  
 COMPQ = 'N'  
 The array Q is unchanged.  
 COMPQ = 'V'  
 The left transformation  $Q$  is accumulated on the array Q.  
 COMPQ = 'I'  
 The array Q is initialized to the identity matrix before the left transformation  $Q$  is accumulated in Q.  
*Constraint:* COMPQ = 'N', 'V' or 'I'.
- 3: COMPZ – CHARACTER(1) *Input*  
*On entry:* specifies the operations to be performed on  $Z$ .  
 COMPZ = 'N'  
 The array Z is unchanged.  
 COMPZ = 'V'  
 The right transformation  $Z$  is accumulated on the array Z.

COMPZ = 'I'

The array  $Z$  is initialized to the identity matrix before the right transformation  $Z$  is accumulated in  $Z$ .

*Constraint:* COMPZ = 'N', 'V' or 'I'.

4: N – INTEGER *Input*

*On entry:*  $n$ , the order of the matrices  $A$ ,  $B$ ,  $Q$  and  $Z$ .

*Constraint:*  $N \geq 0$ .

5: ILO – INTEGER *Input*

6: IHI – INTEGER *Input*

*On entry:* the indices  $i_{lo}$  and  $i_{hi}$ , respectively which define the upper triangular parts of  $A$ . The submatrices  $A(1 : i_{lo} - 1, 1 : i_{lo} - 1)$  and  $A(i_{hi} + 1 : n, i_{hi} + 1 : n)$  are then upper triangular. These parameters are provided by F08WHF (DGGBAL) if the matrix pair was previously balanced; otherwise,  $ILO = 1$  and  $IHI = N$ .

*Constraints:*

if  $N > 0$ ,  $1 \leq ILO \leq IHI \leq N$ ;  
if  $N = 0$ ,  $ILO = 1$  and  $IHI = 0$ .

7: A(LDA,\*) – REAL (KIND=nag\_wp) array *Input/Output*

**Note:** the second dimension of the array  $A$  must be at least  $\max(1, N)$ .

*On entry:* the  $n$  by  $n$  upper Hessenberg matrix  $A$ . The elements below the first subdiagonal must be set to zero.

*On exit:* if JOB = 'S', the matrix pair  $(A, B)$  will be simultaneously reduced to generalized Schur form.

If JOB = 'E', the 1 by 1 and 2 by 2 diagonal blocks of the matrix pair  $(A, B)$  will give generalized eigenvalues but the remaining elements will be irrelevant.

8: LDA – INTEGER *Input*

*On entry:* the first dimension of the array  $A$  as declared in the (sub)program from which F08XEF (DHGEQZ) is called.

*Constraint:*  $LDA \geq \max(1, N)$ .

9: B(LDB,\*) – REAL (KIND=nag\_wp) array *Input/Output*

**Note:** the second dimension of the array  $B$  must be at least  $\max(1, N)$ .

*On entry:* the  $n$  by  $n$  upper triangular matrix  $B$ . The elements below the diagonal must be zero.

*On exit:* if JOB = 'S', the matrix pair  $(A, B)$  will be simultaneously reduced to generalized Schur form.

If JOB = 'E', the 1 by 1 and 2 by 2 diagonal blocks of the matrix pair  $(A, B)$  will give generalized eigenvalues but the remaining elements will be irrelevant.

10: LDB – INTEGER *Input*

*On entry:* the first dimension of the array  $B$  as declared in the (sub)program from which F08XEF (DHGEQZ) is called.

*Constraint:*  $LDB \geq \max(1, N)$ .

11: ALPHAR(N) – REAL (KIND=nag\_wp) array *Output*

*On exit:* the real parts of  $\alpha_j$ , for  $j = 1, 2, \dots, n$ .

- 12: ALPHAI(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the imaginary parts of  $\alpha_j$ , for  $j = 1, 2, \dots, n$ .
- 13: BETA(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:*  $\beta_j$ , for  $j = 1, 2, \dots, n$ .
- 14: Q(LDQ,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array Q must be at least  $\max(1, N)$  if COMPQ = 'V' or 'I' and at least 1 if COMPQ = 'N'.  
*On entry:* if COMPQ = 'V', the matrix  $Q_0$ . The matrix  $Q_0$  is usually the matrix  $Q$  returned by F08WEF (DGGHRD).  
 If COMPQ = 'N', Q is not referenced.  
*On exit:* if COMPQ = 'V', Q contains the matrix product  $QQ_0$ .  
 If COMPQ = 'I', Q contains the transformation matrix  $Q$ .
- 15: LDQ – INTEGER *Input*  
*On entry:* the first dimension of the array Q as declared in the (sub)program from which F08XEF (DHGEQZ) is called.  
*Constraints:*  
     if COMPQ = 'V' or 'I',  $LDQ \geq N$ ;  
     if COMPQ = 'N',  $LDQ \geq 1$ .
- 16: Z(LDZ,\*) – REAL (KIND=nag\_wp) array *Input/Output*  
**Note:** the second dimension of the array Z must be at least  $\max(1, N)$  if COMPZ = 'V' or 'I' and at least 1 if COMPZ = 'N'.  
*On entry:* if COMPZ = 'V', the matrix  $Z_0$ . The matrix  $Z_0$  is usually the matrix  $Z$  returned by F08WEF (DGGHRD).  
 If COMPZ = 'N', Z is not referenced.  
*On exit:* if COMPZ = 'V', Z contains the matrix product  $ZZ_0$ .  
 If COMPZ = 'I', Z contains the transformation matrix  $Z$ .
- 17: LDZ – INTEGER *Input*  
*On entry:* the first dimension of the array Z as declared in the (sub)program from which F08XEF (DHGEQZ) is called.  
*Constraints:*  
     if COMPZ = 'V' or 'I',  $LDZ \geq N$ ;  
     if COMPZ = 'N',  $LDZ \geq 1$ .
- 18: WORK(max(1,LWORK)) – REAL (KIND=nag\_wp) array *Workspace*  
*On exit:* if INFO = 0, WORK(1) contains the minimum value of LWORK required for optimal performance.
- 19: LWORK – INTEGER *Input*  
*On entry:* the dimension of the array WORK as declared in the (sub)program from which F08XEF (DHGEQZ) is called.

If  $LWORK = -1$ , a workspace query is assumed; the routine only calculates the minimum size of the  $WORK$  array, returns this value as the first entry of the  $WORK$  array, and no error message related to  $LWORK$  is issued.

*Constraint:*  $LWORK \geq \max(1, N)$  or  $LWORK = -1$ .

20: INFO – INTEGER

*Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If  $INFO = -i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If  $1 \leq INFO \leq N$ , the  $QZ$  iteration did not converge and the matrix pair  $(A, B)$  is not in the generalized Schur form at exit. However, if  $INFO < N$ , then the computed  $\alpha_i$  and  $\beta_i$  should be correct for  $i = INFO + 1, \dots, N$ .

If  $N + 1 \leq INFO \leq 2 \times N$ , the computation of shifts failed and the matrix pair  $(A, B)$  is not in the generalized Schur form at exit. However, if  $INFO < 2 \times N$ , then the computed  $\alpha_i$  and  $\beta_i$  should be correct for  $i = INFO - N + 1, \dots, N$ .

If  $INFO > 2 \times N$ , then an unexpected Library error has occurred. Please contact NAG with details of your program.

## 7 Accuracy

Please consult Section 4.11 of the LAPACK Users' Guide (see Anderson *et al.* (1999)) and Chapter 6 of Stewart and Sun (1990), for more information.

## 8 Parallelism and Performance

F08XEF (DHGEQZ) is not threaded by NAG in any implementation.

F08XEF (DHGEQZ) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

F08XEF (DHGEQZ) is the fifth step in the solution of the real generalized eigenvalue problem and is called after F08WEF (DGGHRD).

The complex analogue of this routine is F08XSF (ZHGEQZ).

## 10 Example

This example computes the  $\alpha$  and  $\beta$  parameters, which defines the generalized eigenvalues, of the matrix pair  $(A, B)$  given by

$$A = \begin{pmatrix} 1.0 & 1.0 & 1.0 & 1.0 & 1.0 \\ 2.0 & 4.0 & 8.0 & 16.0 & 32.0 \\ 3.0 & 9.0 & 27.0 & 81.0 & 243.0 \\ 4.0 & 16.0 & 64.0 & 256.0 & 1024.0 \\ 5.0 & 25.0 & 125.0 & 625.0 & 3125.0 \end{pmatrix}$$

$$B = \begin{pmatrix} 1.0 & 2.0 & 3.0 & 4.0 & 5.0 \\ 1.0 & 4.0 & 9.0 & 16.0 & 25.0 \\ 1.0 & 8.0 & 27.0 & 64.0 & 125.0 \\ 1.0 & 16.0 & 81.0 & 256.0 & 625.0 \\ 1.0 & 32.0 & 243.0 & 1024.0 & 3125.0 \end{pmatrix}.$$

This requires calls to five routines: F08WHF (DGGBAL) to balance the matrix, F08AEF (DGEQRF) to perform the  $QR$  factorization of  $B$ , F08AGF (DORMQR) to apply  $Q$  to  $A$ , F08WEF (DGGHRD) to reduce the matrix pair to the generalized Hessenberg form and F08XEF (DHGEQZ) to compute the eigenvalues using the  $QZ$  algorithm.

### 10.1 Program Text

```

Program f08xefe

!      F08XEF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: dgeqrf, dggbal, dgghrd, dhgeqz, dormqr, nag_wp, &
                        x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter      :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                  :: i, ifail, ihi, ilo, info, irows,      &
                        jwork, lda, ldb, ldq, ldz, lwork, n
Character (1)            :: compq, compz, job
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: a(:, :), alphai(:), alphas(:), &
                        b(:, :), beta(:), lscale(:), q(:, :), &
                        rscale(:), tau(:), work(:), z(:, :)
!      .. Intrinsic Procedures ..
Intrinsic                :: nint
!      .. Executable Statements ..
Write (nout,*) 'F08XEF Example Program Results'
Flush (nout)

!      Skip heading in data file

Read (nin,*)
Read (nin,*) n
ldq = 1
ldz = 1
lda = n
ldb = n
lwork = 6*n
Allocate (alphai(n), alphas(n), beta(n), a(lda, n), lscale(n), q(ldq, ldq), &
          rscale(n), b(ldb, n), tau(n), work(lwork), z(ldz, ldz))

!      READ matrix A from data file
Read (nin,*) (a(i, 1:n), i=1, n)

```

```

!   READ matrix B from data file
Read (nin,*)(b(i,1:n),i=1,n)

!   Balance matrix pair (A,B)
job = 'B'

!   The NAG name equivalent of dggbal is f08whf
Call dggbal(job,n,a,lda,b,ldb,ilo,ihi,lscale,rscale,work,info)

!   Matrix A after balancing

!   ifail: behaviour on error exit
!           =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04caf('General',' ',n,n,a,lda,'Matrix A after balancing',ifail)

Write (nout,*)
Flush (nout)

!   Matrix B after balancing

ifail = 0
Call x04caf('General',' ',n,n,b,ldb,'Matrix B after balancing',ifail)

Write (nout,*)
Flush (nout)

!   Reduce B to triangular form using QR
irows = ihi + 1 - ilo

!   The NAG name equivalent of dgeqrf is f08aef
Call dgeqrf(irows,irows,b(ilo,ilo),ldb,tau,work,lwork,info)

!   Apply the orthogonal transformation to matrix A
!   The NAG name equivalent of dormqr is f08agf
Call dormqr('L','T',irows,irows,irows,b(ilo,ilo),ldb,tau,a(ilo,ilo),lda, &
work,lwork,info)

!   Compute the generalized Hessenberg form of (A,B) -> (H,T)
compq = 'N'
compz = 'N'

!   The NAG name equivalent of dgghrd is f08wef
Call dgghrd(compq,compz,irows,1,irows,a(ilo,ilo),lda,b(ilo,ilo),ldb,q, &
ldq,z,ldz,info)

!   Matrix A (H) in generalized Hessenberg form.

ifail = 0
Call x04caf('General',' ',n,n,a,lda,'Matrix A in Hessenberg form',ifail)

Write (nout,*)
Flush (nout)

!   Matrix B (T) in generalized Hessenberg form.

ifail = 0
Call x04caf('General',' ',n,n,b,ldb,'Matrix B is triangular',ifail)

!   Routine DHGEQZ
!   Workspace query: jwork = -1

jwork = -1
job = 'E'

!   The NAG name equivalent of dhgeqz is f08xef
Call dhgeqz(job,compq,compz,n,ilo,ihi,a,lda,b,ldb,alphar,alphai,beta,q, &
ldq,z,ldz,work,jwork,info)

Write (nout,*)
Write (nout,99999) nint(work(1))

```

```

Write (nout,99998) lwork
Write (nout,*)

!   Compute the generalized Schur form
!   if the workspace lwork is adequate

If (nint(work(1))<=lwork) Then

!   The NAG name equivalent of dhgeqz is f08xef
Call dhgeqz(job,compq,compz,n,ilo,ihi,a,lda,b,ldb,alphan,alphai,beta, &
  q,ldq,z,ldz,work,lwork,info)

!   Print the generalized eigenvalues

Write (nout,99997)

Do i = 1, n
  If (beta(i)/=0.0E0_nag_wp) Then
    Write (nout,99996) i, '(' , alphan(i)/beta(i), ', ', &
      alphai(i)/beta(i), ', )'
  Else
    Write (nout,99994) i
  End If
End Do
Else
  Write (nout,99995)
End If

99999 Format (1X,'Minimal required LWORK = ',I6)
99998 Format (1X,'Actual value of LWORK = ',I6)
99997 Format (1X,'Generalized eigenvalues')
99996 Format (1X,I4,5X,A,F7.3,A,F7.3,A)
99995 Format (1X,'Insufficient workspace allocated for call to F08XEF/DHGEQZ')
99994 Format (1X,I4,'Eigenvalue is infinite')
End Program f08xefe

```

## 10.2 Program Data

F08XEF Example Program Data

5					:Value of N
1.00	1.00	1.00	1.00	1.00	
2.00	4.00	8.00	16.00	32.00	
3.00	9.00	27.00	81.00	243.00	
4.00	16.00	64.00	256.00	1024.00	
5.00	25.00	125.00	625.00	3125.00	:End of matrix A
1.00	2.00	3.00	4.00	5.00	
1.00	4.00	9.00	16.00	25.00	
1.00	8.00	27.00	64.00	125.00	
1.00	16.00	81.00	256.00	625.00	
1.00	32.00	243.00	1024.00	3125.00	:End of matrix B

## 10.3 Program Results

F08XEF Example Program Results

Matrix A after balancing

	1	2	3	4	5
1	1.0000	1.0000	0.1000	0.1000	0.1000
2	2.0000	4.0000	0.8000	1.6000	3.2000
3	0.3000	0.9000	0.2700	0.8100	2.4300
4	0.4000	1.6000	0.6400	2.5600	10.2400
5	0.5000	2.5000	1.2500	6.2500	31.2500

Matrix B after balancing

	1	2	3	4	5
1	1.0000	2.0000	0.3000	0.4000	0.5000
2	1.0000	4.0000	0.9000	1.6000	2.5000
3	0.1000	0.8000	0.2700	0.6400	1.2500
4	0.1000	1.6000	0.8100	2.5600	6.2500
5	0.1000	3.2000	2.4300	10.2400	31.2500



Matrix A in Hessenberg form

	1	2	3	4	5
1					
2	-2.1898				
3	-0.8395	-0.3181			
4	0.0000	-0.0426	2.0547		
5	0.0000	-0.2846	-1.0101	4.7371	
6	0.0000	0.0000	0.0376	7.5194	-4.6249
7	0.0000	0.0000	0.0000	-7.5927	-17.1850
8	0.0000	0.0000	0.0000	1.4070	26.4499
9	0.0000	0.0000	0.0000	0.3813	-3.3643
10					-0.9937

Matrix B is triangular

	1	2	3	4	5
1					
2	-1.4248				
3	0.0000	-0.3476			
4	0.0000	-0.0782	2.1175		
5	0.0000	0.0000	0.1189	5.5813	
6	0.0000	0.0000	1.0021	-10.9356	-3.9269
7	0.0000	0.0000	0.0000	0.5820	-15.2928
8	0.0000	0.0000	0.0000	0.0000	26.5971
9	0.0000	0.0000	0.0000	0.0000	-0.0730
10	0.0000	0.0000	0.0000	0.0000	0.5321

Minimal required LWORK = 5

Actual value of LWORK = 30

Generalized eigenvalues

1	( -2.437, 0.000)
2	( 0.607, 0.795)
3	( 0.607, -0.795)
4	( 1.000, 0.000)
5	( -0.410, 0.000)

---